

**Universidade Federal de Campina Grande
Centro de Ciências e Tecnologia
Curso de Engenharia Elétrica
Departamento de Engenharia Elétrica**

**Relatório da Disciplina Projeto de Engenharia
Elétrica
Área: Controle e Automação**

**Discente: Lucimar Martins de Oliveira
Orientador: Raimundo Silvério Freire**

**Campina Grande –PB
2004**

**Universidade Federal de Campina Grande
Centro de Ciências e Tecnologia
Curso de Engenharia Elétrica
Departamento de Engenharia Elétrica**

**Relatório da Disciplina Projeto de Engenharia
Elétrica
Área: Controle e Automação**

**Projeto para Controle e Supervisão de Níveis
utilizando um CLP Simatic S7 200 Siemens**

**Discente: Lucimar Martins de Oliveira
Orientador: Raimundo Silvério Freire**

**Campina Grande –PB
2004**



0482p Oliveira, Lucimar Martins de.
Projeto para controle e supervisão de níveis
utilizando um CLP Simatic S7-200 Siemens. / Lucimar
Martins de Oliveira. - Campina Grande - PB: [s.n], 2004.

38 f.

Orientador: Professor Dr. Raimundo Carlos Silvério
Freire.

Trabalho de Conclusão de Curso - Monografia; (Curso
de Bacharelado em Engenharia Elétrica) - Universidade
Federal de Campina Grande; Centro de Engenharia Elétrica
e Informática.

1. Controlador Lógico Programável - CLP. 2. CLP
Simatic S7-200 Siemens. 3. Programa de controle e
supervisão de nível. 4. Automação industrial. 5.
Interface homem máquina. 6. Ambiente de programação Step
7 Micro/Win. I. Freire, Raimundo Carlos Silvério. II.
Título.

CDU:681.5(043.1)

Elaboração da Ficha Catalográfica:

Johnny Rodrigues Barbosa
Bibliotecário-Documentalista
CRB-15/626

Sumário

1. Resumo	2
2. Introdução ao Controlador Lógico Programáveis	3
2.1 Estrutura básica dos CLP's	4
2.2 Vantagens do uso de CLP's	6
2.3 Interface Homem Máquina- IHM	7
3. CLP Simatic S7-200 - Siemens	8
3.1 Interface Homem-Máquina TD-200	9
3.2 Ambiente de Programação: Step 7 Micro/Win	9
4. Projeto de Controle e Supervisão de Nível	11
5. Conclusão	22
6. Referências Bibliográficas	23
7. Anexos	24

1. Resumo

Este trabalho tem como objetivo o desenvolvimento de um programa de supervisão e controle de nível de óleo em reservatórios de um sistema de filtragem de um Laminador. Para o desenvolvimento deste sistema foi proposta a utilização de um Controlador Lógico Programável, ferramenta fundamental na área da Automação Industrial.

Este trabalho apresenta uma breve introdução aos Controladores Lógico Programável, com uma descrição sucinta das características, vantagens oferecidas, estrutura e descrição do funcionamento. Em seguida é possível verificar uma introdução as IHM - Interface Homem Máquina que em conjunto com os CLP's têm sido de fundamental importância para os sistemas que necessitam ser supervisionados.

O CLP Simatic S7-200 – Siemens e a Interface Homem Máquina TD-200 utilizada no projeto são descritos.

Em seguida são descritos os procedimentos utilizados para o desenvolvimento do programa de controle e supervisão de níveis.

O Anexo apresenta a documentação elaborada para o sistema desenvolvido.

2. Introdução ao Controlador Lógico Programáveis

O objetivo inicial do projeto de desenvolvimento do primeiro controlador programável era eliminar o alto custo associado com os sistemas controlados a relés. Para esse desenvolvimento foram especificadas algumas características que o sistema deveria possuir, tais como:

- **flexibilidade de um computador**
- **suportar o ambiente industrial**
- **facilidade de programação**
- **manutenção facilitada**
- **permitir expansão**

A NEMA - National Electrical Manufacturers Association definiu, em 1978, um padrão para controladores programáveis como sendo "um aparelho eletrônico digital que usa uma memória programável para armazenamento interno de instruções para implementar funções específicas tais como lógica, seqüenciamento, temporização, contagem e operações aritméticas, para controlar máquinas ou processos através de módulos de entradas/saídas analógicos ou digitais".

Os primeiros controladores possuíam baixa capacidade de processamento, isto restringia suas aplicações a algumas máquinas e pequenos processos. Com a introdução da tecnologia de microprocessadores, os controladores passaram a ter uma grande capacidade de processamento e alta flexibilidade de programação e expansão, além da capacidade de operar com números, capacidade de comunicação com computadores, aquisição e manipulação de dados, aumento na capacidade de memória e de entradas e saídas.

Atualmente, existem vários tipos de controladores, desde pequena capacidade até os mais sofisticados realizando operações que antes eram consideradas específicas para computadores.

2.1 Estrutura básica dos CLP's

Um controlador programável, independente do tamanho, custo ou complexidade, consiste de

cinco elementos básicos:

- Processador;
- Memória;
- Sistema de entradas/saídas;
- Fonte de alimentação;
- Terminal de programação.

A três partes principais (processador, memória e fonte de alimentação) formam o que chamamos de CPU - Unidade Central de Processamento.

O Processador lê dados de entrada de vários dispositivos, executa o programa do usuário armazenado na memória e envia dados de saída para comandar os dispositivos de controle. Este processo de leitura das entradas, execução do programa e controle das saída é feito de uma forma contínua e é chamado de ciclo de varredura.

O sistema de entrada/saída forma a interface pelo qual os dispositivos de campo, são conectados ao controlador. Esses dispositivos de campo são sensores, botoeiras, lâmpadas sinalizadoras. O propósito desta interface é condicionar os vários sinais recebidos ou enviados ao mundo externo.

Sinais provenientes de sensores tais como, chaves limites, sensores analógicos, chaves seletoras são conectados aos terminais dos módulos de entrada. Dispositivos que devem ser controlados, como válvulas solenóides, lâmpadas sinalizadoras e outros, são conectados aos terminais dos módulos de saída.

A fonte de alimentação fornece todas as tensões necessárias para a devida operação do CLP e da interface dos módulos de entrada e saída.

A Fig.1 mostra a Estrutura básica do CLP.




Fig.1 Estrutura básica do CLP.

O Controlador Programável tem uma forma particular de trabalhar que caracteriza o seu funcionamento. O controlador opera executando uma seqüência de atividades definidas e controladas pelo programa desenvolvido por um programador. Este modo de operação ocorre em um ciclo, chamado de Ciclo de Varredura ("Scan"), que consiste em :

- leitura das entradas externas;
- execução da lógica programada;
- atualização das saídas externas.

Na fase de leitura das entradas, o Processador endereça o sistema de E/S, obtém os estados dos dispositivos que estão conectados, e armazena estas informações na forma de bits "1" ou "0", dependendo do estado obtido. A

região da memória utilizada para armazenar estas informações é chamada de Tabela Imagem das Entradas.

Na fase de execução da lógica programada pelo usuário, a CPU consulta a Tabela de Imagem de Entrada para obter os estados dos dispositivos. Nesta fase, os resultados das lógicas programadas cujas saídas tenham um ponto correspondente no módulo de saída são armazenados em uma área de memória que é chamada de Tabela Imagem das Saídas

Na fase de atualização de saídas, a CPU executa uma varredura na tabela TIS e atualiza as saídas externas, endereçando o Sistema de E/S para atualizar o estado dos dispositivos externos de acordo com o resultado da lógica programada. A seguir, o ciclo é reiniciado e a operação continua enquanto se mantém o controlador no modo de execução.

2.2 Vantagens do uso de CLP's

O rápido desenvolvimento e crescimento da competição fizeram o custo do controlador programável cair significativamente, a ponto de que o estudo de CLP's versus relés, no ponto de vista de custo não ser mais válido. As características, descritas abaixo, mostram as vantagens do uso de CLP's:

- Necessidade de flexibilidade de mudanças na lógica de controle;
- Necessidade de alta confiabilidade;
- Espaço físico disponível pequeno;
- Expansão de entradas e saídas;
- Modificação rápida;
- Lógicas similares em várias máquinas;
- Comunicação com computadores em níveis superiores.

2.3 Interface Homem Máquina- IHM

Com a grande difusão do CLP nesta última década, surgiram também novas necessidades do usuário, como por exemplo, a possibilidade de modificar certos parâmetros dentro de um programa, tais como: o preset de um temporizador ou contador, o preset de uma temperatura desejada, a indicação de uma nova velocidade para um servo-motor; entre outros, sem a necessidade de conectar-se a um computador para a realização desta tarefa.

Diante desta necessidade, surgiram as chamadas interfaces homem-máquina, também conhecidas como IHM's.

O princípio de funcionamento das IHM's consiste em pré-programar mensagens, onde cada uma possui um endereço de memória. Esses endereços serão utilizadas nas lógicas para designar os momentos que estas mensagens devem ser apresentadas na IHM.

As teclas de função podem funcionar como botões de comando para acionar qualquer elemento no CLP. A cada tecla, assim como nas mensagens, é atribuído um endereço de memória do CLP .

Ao acionar uma tecla da IHM é possível "setar" um *bit* que poderá ser utilizado pelo programa do CLP para acionar, por exemplo, uma saída que liga um motor de uma bomba ou uma lâmpada de sinalização.

Tal possibilidade é muito vantajosa, pois, além de eliminar a necessidade de se ter um painel convencional de grandes dimensões, com botões e lâmpadas de controle, é possível enviar mensagens para um mostrador que poderão auxiliar no processo. A Fig 2 mostra um TD 200 IHM da Siemens.




Fig 2- IHM TD-200 Siemens

Caso o usuário queira modificar um parâmetro qualquer, como por exemplo, o “preset” de um temporizador, é possível habilitar o teclado numérico frontal, juntamente com a tecla “enter” para que ele possa acessar estes dados.

Existem também as IHM’s com tecnologia *Touch-Screem* (toque de tela) onde se pode introduzir dados ou realizar modificações através de uma tecla Keypad, que quando tocada mostra um teclado numérico para que seja possível digitar o valor.

3. CLP Simatic S7-200 – Siemens

A família de controladores programáveis S7-200 foi desenvolvida para o controle de uma ampla gama de aplicações de controle e automação.

O S7-200 é pequeno e compacto ideal para as aplicações onde o espaço disponível é crítico. Ele também é rápido, oferecendo um excelente comportamento em tempo real, garantindo maior qualidade, eficiência e confiabilidade ao processo.

A família S7-200 é composta de CPU’s com diferentes níveis de memória e diferentes números de entradas e saídas integradas.

A comunicação do S7-200 com outros CLP’S, computadores ou terminal de programação pode ser realizada por meio das interfaces Integradas

padrão RS485 e por meio módulos de expansão específicos. Também é possível a comunicação via modem, PROFIBUS-DP, AS-Interface e Ethernet.

A interface RS485 integrada possibilita a comunicação com um máximo de 126 estações ou nós, sendo possível, sem qualquer problema, ligar em rede dispositivos de programação. Em redes S7-200 puras é utilizado o protocolo PPI integrado. Numa rede composta por componentes SIMATIC S7-300, S7-400 e IHM, as CPUs S7-200 são ligadas como escravas MPI

3.1 Interface Homem-Máquina TD-200

Os CLP's da linha S7-200 podem usar qualquer painel de operação da linha Simatic HMI (Siemens). Porém, para manter sua característica de baixo custo, foi desenvolvido uma linha de painéis de operação específicos para o uso com o S7-200, os Micro Painéis. Essa linha é constituída de dois modelos:


- TD-200: display de texto de duas linhas com teclas programáveis, velocidade de até 187,5 Kbps ;
- TD070: painel gráfico com tela de 5,7" do tipo "touch-screen" com elevado contraste

3.2 Ambiente de Programação: Step 7 Micro/Win

O Step-7 Micro/Win é o software da Siemens utilizado para o desenvolvimento de programas para os CLP da família S7-200. Esse software possui a mesma configuração de menus que outros programas para Windows. A Fig.3 é apresenta a tela de abertura do software.

Essa tela possui uma barra de comandos e ferramentas com os comandos característicos do Windows como: novo arquivo, abrir arquivo, salvar,

imprimir, recortar, copiar e colar. Nesta barra também contém outros ícones específicos que são detalhados a seguir:



Compile - Compilar o programa



Upload - Ler o programa do CLP



Download – armazenar o programa no CLP



Run - Executar o programa



Stop - Parar a execução do programa




Fig.3 - Tela inicial do Step7- Micro/Win

No Step 7- Micro/Win é possível utilizar três tipos de programação, são elas:

- Programação SFC — Diagramas de Funções Seqüências

- Programação STL (Lista de Instruções) — linguagem que permite criar programas utilizando instruções de mnemônicos.
- Programação Ladder (Linguagem de relés ou diagramas de contatos) — Como o próprio nome sugere as instruções básicas se originaram no diagrama eletromecânico, cujo elemento principal de controle é o relé, especificamente sua bobina e seus contatos. Por ser a primeira linguagem utilizada pelos fabricantes é muito difundida e recebeu vários nomes desde sua criação, entre eles: diagrama de escada ("ladder"), diagrama de contatos e linguagem de contatos.

4. Projeto de Controle e Supervisão de Nível


Como uma aplicação para o PLC S7-200 da Siemes foi proposto desenvolver um programa para monitoramento e controle do nível de três reservatórios de óleo de um sistema de filtragem de óleo de um Laminador. Esses tanques são denominados de Tanque de Lama, Tanque Preparação e Tanque de Lubrificação.

Instalados em cada um dos tanques, há sensores ultra-sônicos do tipo Probe que informam, por meio de um mostrador a porcentagem de óleo existente nos tanques. Esses dados são obtidos através de uma escala de corrente, com uma variação de 4 a 20 mA, ou seja, é realizada uma leitura em corrente proporcional ao nível de óleo existente, sendo que 4 mA corresponde ao tanque está cheio e 20 mA corresponde a tanque vazio.

O projeto foi sugerido para eliminar a lógica utilizando relés e temporizadores que apresentava dificuldades de manutenção devido aos equipamentos utilizados serem antigos e não possuem peças de reposição.

Conhecendo-se a capacidade volumétrica de cada um dos tanques, foi desenvolvida uma escala para converter a leitura analógica realizada pelo CLP em valores que indicassem para cada tanque o nível atual em litros e a porcentagem da capacidade. Além da capacidade volumétrica, foi necessário

Fluxogramas do Programa de Controle e Supervisão de Nível



conhecer o funcionamento de cada tanque, ou seja, os níveis considerados normais, críticos e de alerta, pois esses dados serão utilizados como referências no programa.

De posse desses dados, o processo de programação foi iniciado. Para tal foi utilizado um CLP S7-200, um simulador de entradas analógicas e digitais, um computador e uma Interface Homem-Máquina TD 200 da Siemens.

O simulador de entradas digitais e analógicas possui quatro potenciômetros que foram utilizados para simular as leituras dos sensores. Esse simulador também possui lâmpadas sinalizadoras ligadas as saídas do CLP que facilitam a visualização das saídas que estão sendo utilizadas pelo programa.


Para o desenvolvimento do programa foi utilizada a linguagem STL (Lista de Instruções), um exemplo desse tipo de programação é mostrado a seguir:

```
LD SM0.0
MOVW AIW4, VW102
MOVR VD130, VD116
MOVR VD134, VD120
```


Para desenvolver o programa foram desenvolvidos fluxogramas. Através da análise desses fluxogramas, mostrados a seguir, pode-se visualizar o desenvolvimento do programa:

Em seguida para cada tanque é realizada a comparação do resultado da com os valores de referência. Os fluxogramas que se seguem apresentam os passos que devem ser seguidos:


Tanque de Preparação



Tanque de Lama



Tanque de Lubrificação dos Redutores



Após a análise dos fluxogramas apresentados foi desenvolvido o código. Esse código implementa as lógicas sugeridas nos fluxogramas.

No programa principal as variáveis são iniciadas. Os valores de referência são armazenados, os valores iniciais são “setados”, e entradas e saídas são “resetados” e as mensagens do menu inicial são habilitadas no mostrador do TD-200. Essa etapa do desenvolvimento pode ser verificada no código a seguir:

// Iniciação das variáveis

```
NETWORK 1           // Tanque de Lama
LD  SM0.1
MOVR 0.0, VD138     // Volume Minimo Tanque de Lama
MOVR 1500.0, VD142  // Volume Minimo Tanque de Lama

NETWORK 2 //Tanque de Preparação
LD  SM0.1
MOVR 0.0, VD130     // Volume Minimo Tanque de Preparação
MOVR 400.0, VD134   // Volume Máximo Tanque de Preparação

NETWORK 2 //Tanque de Lubrificação
LD  SM0.1
MOVR 0.0, VD146     // Volume Minimo Tanque de Lubrificação
MOVR 400.0, VD150   // Volume Máximo Tanque de Lubrificação

NETWORK 3
LD  SM0.1
R   V12.0, 16       // Reset Variáveis
R   Q0.0, 8         // Reset Saídas
R   Q1.0, 1         // Reset as Saídas
R   M11.0, 16      // Reset Flags
R   M0.0, 8        // Reset Flags
R   M10.0, 16     // Reset Flags
S   V14.0, 3       // Seta o menu Principal

NETWORK 4
LD  SM0.1
MOVR 92.0, VD430    // Referência Nível alto tanque de preparação
MOVR 95.0, VD450    // Referência Nível bomba tanque de preparação
MOVR 10.0, VD470    // Referência Nível baixo tanque de preparação
```

MOVR 45.0, VD490 // Referência Nível aviso tanque de preparação

NETWORK 5

LD SM0.1

MOVR 90.0, VD670 // Referência Nível alto tanque de lama

MOVR 20.0, VD690 // Referência Nível baixo tanque de lama

MOVR 80.0, VD710 // Referência Nível aviso tanque de lama

NETWORK 6

LD SM0.1

MOVR 40.0, VD910 // Referência Nível baixo tanque lubrificação

MOVR 42.0, VD930 // Referência Nível de aviso lubrificação

MOVW +1000, VW160

Na segunda etapa é realizada a leitura dos sensores. O valor de cada sensor é armazenado em uma posição de memória específica e, em seguida são convertidos para litros e para percentual de litros, por meio da sub-rotina Escala, onde o cálculo demonstrado a seguir é implementado:

Y=Leitura desejada

X=Entrada analógica (Leitura do Sensor)

Vol. Máx=Volume Máximo do Tanque (Valor de referência)

Vol. Min= Volume Mínimo do Tanque (Valor de referência)

$Y = -[(Vol.Máx - Vol.Min)/24600] * X + 30000 * [(Vol.Máx$

$Vol.Min)/24600] + Vol.Min$

$Y\% = (Y / Vol. Máx) * 100$

O código da sub-rotina Escala é o seguinte:

LD SM0.0

MOVW +0, VW100

DTR VD100, VD104

MOVR VD120, AC0 //Armazena o Volume máximo no Acumulador

-R VD116, AC0 //Subtrai do volume mínimo

/R -24400.0, AC0 //Divide o resultado por -24600, tem-se o coefic.angular

*R AC0, AC1 //Multiplica o coefic.angular pela entrada analógica

```

*R  30800.0, AC0    //Multiplica o coeficiente angular pelo maior valor nalógico
                        //ido pelo Sensor
-R  AC0, AC1        //Subtrai o resultado das duas multiplicações
+R  VD116, AC1      //Soma o valor de volume mínimo do tanque
MOVR AC1, VD112     //Armazena este resultado em uma Posição da

```

memória

```

/R  VD120, AC1      //Divide este resultado pelo Volume máximo do

```

Tanque

```

*R  100.0, AC1      //Multiplica por 100%
MOVR AC1, VD124     //Armazena o resultado em percentagem em uma

```

posição de memória

Após a conversão dos valores, realizadas pela sub-rotina Escala, os resultados são comparados com as referências. De acordo com os resultados dessa s comparações as saídas e mensagens serão habilitadas.

Sempre que uma mensagem de nível crítico é habilitada no mostrador do TD-200, uma saída é habilitada e essa saída aciona uma sirene. A lógica que implementa o acionamento dessa sirene é mostrada a seguir:

```

NETWORK 27    // ALARME SONORO
LD  M10.2      // Se o nível do tanq. de Preparação é critico
O  M11.0       // Ou se o nível do tanque de Lama é critico
O  M12.0       // Ou se o nível do tanque de Lubrificação é critico
=  Q0.6        // Alarme sonoro

```

Para programar o TD-200 é utilizado o Step7-Micro/Win. Esse software possui uma ferramenta, TD-200 Wizard que é acessada por meio do Menu Tools da barra de Ferramenta, como mostra a Fig.4.




Fig 4 - Ferramenta Td200 Wizard

Essa ferramenta permite ao programador editar as mensagens, como mostra a Fig.5, que serão utilizadas no programa.




Fig 5 - Tela para edição de mensagens

Para cada mensagem é atribuído um endereço de memória e, é este endereço que será habilitado no programa quando for desejado que uma mensagem apareça na IHM. A Fig.6 mostra um exemplo de uma mensagem no mostrador do TD-200.



Fig 6 - TD-200

As mensagens apresentadas no TD200 estão divididas em quatro menus:

- O menu principal acionado por meio da tecla F4 do TD 200;
- O menu Tanque de Preparação acionado por meio da tecla F1 do TD 200;

- O menu Tanque de Lama acionado por meio da tecla F2 do TD200;
- O menu Tanque de Lubrificação acionado por meio da tecla F3 do TD 200;

Cada uma dessas teclas possui um endereço de memória associado, que é o endereço utilizado no programa para implementação das lógicas relacionadas com esses menus.

Após a conclusão dos trabalhos de programação foi realizada a montagem do painel, para tal foram utilizados:

- **PLC S7 200 - Siemens CPU 214**
- **Fonte 24v**
- **Foto acopladores**
- **TD200 – Interface Homem Máquina-Siemens**

Em seguida foi elaborada a documentação do sistema, que consiste dos seguintes documentos:

- Descrição Funcional do programa;
- Cópia do programa comentado em linguagem STL e em Linguagem Ladder;
- Diagrama elétrico das ligações do painel;
- Manual dos Sensores;
- Procedimento de como re-carregar o programa;
- Procedimento de Backup;

5. Conclusão

O estudo de uma ferramenta como o Controlador Lógico Programável e a possibilidade de implementação de sistema real utilizando uma tecnologia utilizada frequentemente no mundo das indústrias como o CLP S7 200 Siemens proporcionou um aprendizado fundamental para o desenvolvimento do engenheiro em aplicações de automação industrial.

No processo de desenvolvimento do projeto foi possível conhecer inúmeras ferramentas que auxiliam no desenvolvimento de projetos de automação, tais como, sensores, atuadores e tecnologias diversas utilizadas para desenvolvimento de sistemas de controle e supervisão.

Todo este conhecimento será de fundamental importância na vida profissional, pois além de adquirir experiência com as tecnologias envolvidas, foi possível observar a necessidade de constante atualização de tecnologias que o profissional deve acompanhar. Pois as indústrias de grande e médio porte buscam nos profissionais conhecimentos que auxiliem na produção no intuito de satisfazer as exigências dos clientes, e estas exigências acompanham as atualizações do mercado de tecnologias.

6. Referências Bibliográficas

S7 – 200 programmable controller system Manual. 6 e. Siemens, 2003.

TD 200 operator interface user manual. 5 e. Siemens, 1999.

7. Anexos

Código do Programa de Supervisão e Controle de Nível

```

NETWORK 1          // Tanque de Lama
LD      SM0.1
MOVR   0.0, VD138          //Volume Minimo Tanque de Lama
MOVR   1500.0, VD142       //Volume Minimo Tanque de Lama

NETWORK 2          //Tanque de Preparação
LD      SM0.1
MOVR   0.0, VD130          //Volume Minimo Tanque de
Preparação
MOVR   400.0, VD134       //Volume Máximo Tanque de
Preparação

NETWORK 3
LD      SM0.1
MOVR   0.0, VD146          //Volume Minimo Tanque de
Lubrificação
MOVR   100.0, VD150       //Volume Máximo Tanque de
Lubrificação

NETWORK 4
LD      SM0.1
R      V12.0, 16          // Reset Variáveis
R      Q0.0, 8            // Reset Saídas
R      Q1.0, 1            // Reset as Saídas
R      M11.0, 16         // Reset Flags e variáveis
R      M0.0, 8           // Reset Flags e variáveis
R      M10.0, 16        // Reset Flags e variáveis
S      V14.0, 3          // Seta o menu Principal

NETWORK 5
LD      SM0.1
MOVR   92.0, VD430        // Referência Nível alto tanque
de preparação
MOVR   95.0, VD450        // Referência Nível bomba tanque
de preparação
MOVR   10.0, VD470        // Referência Nível baixo tanque
de preparação
MOVR   45.0, VD490        // Referência Nível aviso tanque
de preparação

NETWORK 6
LD      SM0.1
MOVR   90.0, VD670        // Referência Nível alto tanque
de lama
MOVR   20.0, VD690        // Referência Nível baixo tanque
de lama
MOVR   80.0, VD710        // Referência Nível aviso tanque
de lama

NETWORK 7
LD      SM0.1
MOVR   40.0, VD910        // Referência Nível baixo tanque
lub. redutores
MOVR   42.0, VD930        // Referência Nível aviso lub.
redutores
MOVW   +1000, VW160

NETWORK 8
LDR>= VD883, VD910       //SISTEMA DE LUBRIFICAÇÃO OK
S      Q0.5, 1

```

```

NETWORK 9      //Leitura da variável analógica - Probe_3- Tanque de
                Preparação
//
//NETWORK COMMENTS
//
LD      SM0.0           // Sempre ativo
MOVW   AIW4, VW102     // Sinal analógico (Leitura do
                sinal do probe 3)
MOVR   VD130, VD116   // Armazena Referência do Tanque
                de Preparação
MOVR   VD134, VD120   // Armazena Referência do Tanque
                de Preparação
CALL   Escala         // Chamada para subrotina Escala
MOVR   VD112, VD403   // Move o valor convertido para a
                mensagem do Td_200/
MOVR   VD124, VD412   // Move o valor convertido para a
                mensagem do Td_200

NETWORK 10     //Leitura da entrada analógica - Probe_2- Tanque de
                Lama
//
//NETWORK COMMENTS
//
LD      SM0.0           // Sempre ativo
MOVW   AIW2, VW102     // Sinal analógico (Leitura do
                sinal do probe 2)
MOVR   VD138, VD116   // Armazena Referência do Tanque
                de Lama
MOVR   VD142, VD120   // Armazena Referência do Tanque
                de Lama
CALL   Escala         // Chamada para subrotina Escala
MOVR   VD112, VD643   // Move o valor convertido para a
                mensagem do Td_200
MOVR   VD124, VD653   // Move o valor convertido para a
                mensagem do Td_200

NETWORK 11     //Leitura da entrada analógica (sensor)- Probe_3-
                Tanque Lub. Redutores
//
//NETWORK COMMENTS
//
LD      SM0.0
MOVW   AIW0, VW102     // Sinal analógico (Leitura do
                sinal do probe 1)
MOVR   VD146, VD116   // Armazena Referência do Tanque
                Lub. Redutores
MOVR   VD150, VD120   // Armazena Referência do Tanque
                Lub. Redutores
CALL   Escala         // Chamada para subrotina Escala
MOVR   VD124, VD883   // Move o valor convertido para a
                mensagem do Td_200

NETWORK 12     // TANQUE DE PREPARAÇÃO EM NÍVEL CRITICO
//
LDR<=  VD412, VD470   // Compara a leitura do sensor
                com a referência
AR>=   VD412, 0.0     // Se o Nível < 10%
=      M10.2         // Energiza a bobina
=      Q0.1          // caso a bobina esteja desernegi
                zada
                // a Saída indica que o nível não
                está abaixo de 10%

NETWORK 13     // NÍVEL TANQUE DE PREPARAÇÃO NIVEL ALTO

```

```

//
LDR>= VD412, VD430 // Compara a leitura do sensor
      com a referência
=      M10.0 // Se Nível >= 92% seta o Flag
=      Q0.2 // e acende o led indicando que
      o nível >= 92% // Saída que indica o nível acima
      de 92%

NETWORK 14 // NÍVEL TANQUE DE PREPARAÇÃO- ACIONAMENTO DA BOMBA
//
LDR>= VD412, VD450 // Compara a leitura do sensor
      com a referência
=      M10.1 // Se o Nível >=95%, seta o
      flag, aciona a Bomba e
=      Q0.3 // acende o Led que indica o
      nível de acionamento da Bomba

NETWORK 15 //TANQUE DE PREPARAÇÃO EM ALERTA
//
LDR<= VD412, VD490 // Compara a leitura do sensor
      com a referência
AR>= VD412, VD470 // Se 10%=< Nível <= 45%
=      M10.3 // Seto o Flag indicando nível
      de alerta
=      Q0.0

NETWORK 16 // TANQUE DE LAMA EM NÍVEL CRITICO
//
LDR>= VD653, VD670 // Compara a leitura do sensor
      com a referência
=      M11.0 // Se Nível >= 90%, seta o flag e
      habilita a
=      Q0.4 // Saída que indica tanque
      acima de 90% de // sua capacidade

NETWORK 17 // TANQUE DE LAMA ABAIXO E 20%
LDR<= VD653, VD690 // Compara o nível
MOVVR VD653, VD565
=      M11.3 // Se nivel<=10%
=      Q0.7 // Habilita saída
=      V12.1

NETWORK 18 // TANQUE DE LAMA EM ALERTA
//
LDR>= VD653, VD710 // Compara o nível do tanque com
      a referência
AR<= VD653, VD670 // Se Tanque de Lama 80%<=Nível<=
      90
=      M11.2 // Seto o Flag
=      Q1.1

NETWORK 19 //TANQUE LUB. REDUTORES- CRITICO
LDR<= VD883, VD910 // Compara o nível do tanque com
      a referência
AR>= VD883, 0.0 // Se 0%<=Nível<= 40%
=      M12.0 // Seto o Flag que indica nível
      crítico
R      Q0.5, 1 // indica alarme para o PLC
      (DESLIGA O LAMINADOR)

NETWORK 20 //Mensagem Aviso Tanque Lub. Redutores
//
LDR<= VD883, VD930 // Compara o nível do tanque com
      a referência

```

```

AR>= VD883, VD910 // Se 40%=<Nível<=42%
= M12.2 // Seta o Flag
= Q1.0

NETWORK 21 //Tanque lub .Redutores nível normal
LDR>= VD883, VD930 //Compara o nível do tanque com a
referência
= M12.1 //Seta o Flag

NETWORK 22 // Chamada para subrotina Td_200
LD SM0.0 // Bit sempre ativo
CALL Td_200 //Chamada para subrotina

NETWORK 23 // PROBE UM APRESNTA DEFEITO
LDW<= AIW4, VW160 // se defeito no probe 1
= M15.0 // ativa o flag de defeito

NETWORK 24 // PROBE 2 APRESENTA DEFEITO
LDW<= AIW2, VW160 // se defeito no probe 2
= M15.1 // ativa o flag de defeito

NETWORK 25 // PROBE 3 APRESENTA DEFEITO
LDW<= AIW0, VW160 // se defeito no probe 3
= M15.2 // ativa o flag de defeito

NETWORK 26 // CHAMADA PARA SUBROTINA DEFEITOS
LD SM0.0 //Bit sempre ativo
CALL Defeitos // Chamada para a subrotina
Defeitos

NETWORK 27 // ALARME SONORO
LD M10.2 // Se o nível do tanq. de
Preparação é critico

O M11.0 // Ou se o nível do tanque de
Lama é critico

O M12.0 // Ou se o nível do tanque de
Lama é critico

= Q0.6 // Alarme sonoro

NETWORK 28 //RESET PELO BOTÃO
LD I0.7 // Quando acionado o Botão Reset
R M0.0, 8 // Reset Flags e variáveis
R M11.0, 16 // Reset Flags e variáveis
R M12.0, 8 // Reset Flags e variáveis
R M10.0, 16 // Reset Flags e variáveis
R V12.0, 16 // Reset Flags e variáveis
R V13.0, 7 // Reset Variáveis
R V14.0, 8 // Reset Flags e variáveis
R Q0.0, 8 // Reset Saídas
R Q1.0, 8 // Reset Saídas

NETWORK 29
LD I0.7 //Botão reset Precionado
ED //na transição negativa
S V14.0, 3 //Sete o menu inicial

```

```

//
//SUBROUTINE COMMENTS

//*****Calculos de converção para escala desejada*****
//*****
// Leitura / Escrita analogica
//Efetua a leitura das entradas analogicas
//Efetua escrita nas saidas analogicas

//NETWORK COMMENTS
//

NETWORK 1
LD      SM0.0
MOVW   +0, VW100
DTR    VD100, VD104

NETWORK 2
LD      SM0.0
MOVR   VD120, AC0           //Armazena o Volume máximo no
    Acumulador
-R     VD116, AC0           //Subtrai do volume mínimo
/R     -24400.0, AC0        //Divide o resultado por -24600
    //O resultado é o coeficiente
    angular da reta
MOVR   VD104, AC1
*R     AC0, AC1             //Multiplica o coeficiente
    angular pela entrada analógica
*R     30800.0, AC0         //Multiplica o coeficiente
    angular pelo maior valor analógico
    //lido pelo Probe
-R     AC0, AC1             //Subtrai o resultado das duas
    multiplicações
+R     VD116, AC1           //Soma o valor de volume mínimo
    do tanque
MOVR   AC1, VD112           //Armazena este resultado em uma
    Posição da memória
/R     VD120, AC1           //Divide este resultado pelo
    Volume máximo do Tanque
*R     100.0, AC1           //Multiplica por 100%
MOVR   AC1, VD124           //Armazena o resultado em
    perctagem em uma posição de memória

//Escala
//Y=Leitura desejada
//X=Entrada analógica
//Vol. Máx=Volume máximo do Tanque
//Vol. Min= Volume Mínimo do Tanque

//Y= -[(Vol.Máx - Vol.Min)/24600]*X+30000*[(Vol.Máx - Vol.Min)/24600]
    +Vol.Min
//Y%=(Y/Vol. Máx)*100

```

```

//subrotina que faz o tratamento de defeitos nos Sensores
//
//<***** Defeitos nos Sensores*****>

NETWORK 1
LD      I0.0                // Entrada que indica defeito no
      Probe 1-Tanque de Preparação
O      M15.0                // Flag que indica erro na
      leitura do Probe 1
R      V12.0, 8             // Reset as mensagens
R      V13.0, 8             // Reset as mensagens
=      M16.0                // Seta o Flag

NETWORK 2
LD      I0.1                // Entrada que indica defeito no
      Probe 2-Tanque de Lama
O      M15.1                // Flag que indica erro na
      leitura do Probe 2
R      V12.0, 8             // Reset as mensagens
R      V13.0, 8             // Reset as mensagens
=      M16.1

NETWORK 3
LD      I0.3                // Entrada que indica defeito no
      Probe 2-Tanque de Lub.Redutorea
O      M15.2                // Flag que indica erro na
      leitura do Probe 3
R      V12.0, 8             // Reset as mensagens
R      V13.0, 8             // Reset as mensagens
=      M16.2

NETWORK 4
LD      M16.0                // Se o Flag está ativo
=      V14.5                // Habilita Mensagem de erro no
      Probe
ED      // Transição negativa (Após o
      defeito ser solucionado )
S      V12.6, 2             // Habilita o menu Principal
R      M16.0, 1             // Reset o Flag

NETWORK 5
LD      M16.1                // Se o Flag está ativo
=      V14.4                // Mensagem de erro do Probe
ED      // Transição negativa (após
      defeito solucionado0
S      V12.6, 2             // Habilita menu Principal
R      M16.1, 1             // Reset o Flag

NETWORK 6
LD      M16.2                // Se o Flag está ativo
=      V14.3                // Mensagem de erro do Probe
ED      // Transição negativa (Após o
      defeito ser solucionado )
S      V12.6, 2             // Habilita o menu Principal
R      M16.2, 1

NETWORK 7
LD      SM0.0                // Bit sempre ativo
CALL   Td_200                // Chama a subrotina Td_200

```

```

//
//SUBROUTINE COMMENTS
//Press F1 for help and example program
//

//Subrotina que habilita as mensagens no diplay no Td_200
//
//<*****Tratamento de mensagens Td_200*****
>
//

NETWORK 1 //Tanque de Preparação (F1)
//NETWORK COMMENTS
//
LD V14.0 // Menu principal
O V14.1 // Menu principal
O V14.2 // Menu principal
A M0.0 // F1- habilita o menu Tanque
dePreparação
R V14.0, 3 // Reset o menu principal
R V13.2, 1 // Reset o menu principal

S V12.3, 3

NETWORK 2
LD V14.0 // Menu principal
O V14.1 // Menu principal
O V14.2 // Menu principal
A M0.1 // F2- Habilita o menu Tanque de
Lama
R V14.0, 3 // Reset o menu principal
S V13.5, 3 // Seta as informações do Tanque
de Lama
R M0.1, 1 // Reset a tecla F2

NETWORK 3 //TANQUE DE LUB. REDUTORES (F3)
LD V14.0 // Menu, principal
O V14.1 // Menu principal
O V14.2 // Menu principal
A M0.2 // F2- Habilita o menu Tanque de
Lama
R V14.0, 3 // Reset o menu principal
S V13.1, 1 // Seta as informações do Tanque
Lub. Redutores
S V13.0, 1 // Seta as informações do Tanque
Lub. Redutores
R M0.2, 1 // Reset a tecla F3

NETWORK 4 // TANQUE DE LAMA- NÍVEL CRITICO
//
LD M11.0 // Indica que o tanq. encontra-se
com o nível superior a 80%
= V13.4 // Habilita mensagem de nível
critico alto
EU // Transição positiva
R V12.0, 8 // Reset as mensagens
R V13.0, 8 // Reset as mensagens
R V14.0, 8 // Reset as mensagens

NETWORK 5 // TANQUE DE LAMA - NÍVEL CRITICO
//
LDN M11.0 // Quando o estado normaliza-se e
A V740.1 // a mensagem é reconhecida (Apert
e ENTER)

```

```

R      V13.4, 1          // Reset a mensagem de estado
      critico
S      V14.0, 3          // Habilita o Menu Tanque de
      Lama
R      V740.1, 1         // Reset ENTER

NETWORK 6      // TANQUE DE LAMA - ALERTA
//
LD      M11.2            // Indica tanque de lama em
      alerta
AN      T38              // Temporizador
=      V13.3            // Habilita mensagem de alerta
EU      // Transição positiva
R      V12.0, 8          // Reset Mensagens
R      V14.0, 8          // Reset Mensagens
R      V13.0, 8          // Reset Mensagens

NETWORK 7      // TANQUE DE LAMA - ALERTA
LD      V13.3            // Mensagem de alerta ativa,
      habilita
TON     T38, +600        // o contador

NETWORK 8      // TANQUE DE LAMA - ALERTA
LD      M11.2            // Indica tanque de lama em
      alerta
A      V780.1            // Posição que indica a tecla
      enter pressionado
EU      // Transição positiva
S      V12.6, 2          // Seta o tanques
R      V780.1, 1         // Reset o ENTER

NETWORK 9      // TANQUE DE LAMA - ALERTA
LDN     M11.2            // Se o Flag não está setado
AN      M11.0            // Se a mensagem de tanque
      critico não está ativa
A      V780.1            // e a tecla enter for pressionad
      a
S      V14.0, 3          // Seta o Menu Principal
R      V12.6, 2
R      V780.1, 1

NETWORK 10     // TANQUE DE PREPARAÇÃO - CRITICO
LD      M10.2            // Indica que o tanq. encontra-se
      com o nível inferior a 10%
=      V12.2            // Habilita mensagem de nível
      critico baixo
EU      // Transição positiva
R      V12.0, 8          // Reset as mensagens
R      V13.0, 6          // Reset as mensagens
R      V14.0, 8          // Reset as mensagens

NETWORK 11     // TANQUE DE PREPARAÇÃO - CRITICO
LDN     M10.2            // Quando o estado normaliza-se e
A      V500.1            // a mensagem é reconhecida
      (Aperte ENTER)
R      V12.2, 1          // Reset a mensagem de estado
      critico
S      V12.6, 2          // Habilita o Menu Tanque de
      Preparação
R      V500.1, 1         // Reset o ENTER

NETWORK 12     // TANQUE DE PREPARAÇÃO - ALERTA
LD      M10.3            // Indica tanque de Preparação em
      alerta
AN      T37              // Temporizador

```

```

=      V12.0          // Habilita mensagem de alerta
EU     // Transição positiva
R      V12.0, 8      // Reset as mensagens
R      V13.0, 8      // Reset as mensagens
R      V14.0, 8      // Reset as mensagens

NETWORK 13          // TANQUE DE PREPARAÇÃO EM ALERTA
LD     V12.0          // Mensagem de alerta quando
      ativa seta
TON    T37, +600     // o contador, que fará a
      mensagem retornar                               // após 2 minutos, caso o problema
      não seja solucionado

NETWORK 14          // TANQUE DE PREPARAÇÃO EM ALERTA
LD     M10.3          // Indica tanque de lama em
      alerta
A      V580.1          // O operador deve reconhecer a
      mensagem através da tecla ENTER
EU     // Após a Transição positiva
S      V12.6, 2      // Seta o menu Tanque de
      Preparação
R      V580.1, 1      // Reset ENTER

NETWORK 15          // TANQUE DE PREPARAÇÃO EM ALERTA
LDN    M10.3          // Flag não está ativo
AN     M10.2          // O tanque não está em nível
      critico
A      V580.1          // a tecla enter foi pressionada
S      V14.0, 3
R      V12.6, 2
R      V580.1, 1      // Reset a tecla enter

NETWORK 16          ///TANQUE DE LUB. REDUTORES - CRITICO
LD     M12.0          // Indica que o tanq. encontra-se
      com o nível inferior a 42%
=      V14.7          // Habilita mensagem de nível
      critico baixo
EU     // Após a transição positiva
R      V12.0, 8      // Reset as mensagens
R      V13.0, 8      // Reset as mensagens
R      V14.0, 8      // Reset as mensagens

NETWORK 17          ///TANQUE DE LUB. REDUTORES - CRITICO
LDN    M12.0          // Quando o estado normaliza-se e
A      V940.1          // a mensagem é reconhecida (Apert
      e ENTER)
R      V14.7, 1      // Reset a mensagem de estado
      critico
S      V13.0, 2      // Habilita o Menu Tanque de
      Lub.Redutores
R      V940.1, 1      // Reset o ENTER

NETWORK 18          //TANQUE DE LUB. REDUTORES - ALERTA
LD     M12.2          // Indica tanque de Lub Redutores
      em alerta
AN     T39            // Temporizador
=      V14.6          // Habilita mensagem de Alerta
EU     // Após a transição positiva
R      V12.0, 8      // Reset as mensagens
R      V13.0, 8      // Reset as mensagens
R      V14.0, 8      // Reset as mensagens

NETWORK 19          //TANQUE DE LUB. REDUTORES - ALERTA
LD     V14.6          // Mensagem de alerta quando

```

```

ativa
TON    T39, +600          // o contador, que fará a
mensagem retornar      // após 2 minutos, caso o problema
                        // não seja solucionado

NETWORK 20      //TANQUE DE LUB. REDUTORES - ALERTA
LD      M12.2          // Indica tanque Lub. Redutores
em alerta
A      V980.1          // O operador deve reconhecer a
mensagem através da tecla ENTER
EU      // Após a transição positiva
S      V12.6, 2        //
R      V980.1, 1       // Reset ENTER

NETWORK 21
LDN     M12.2          // Flag não está ativo (Tanque
não está em nível alerta)
AN      M12.0          // o Flag não está ativo (tanque
não está em nível crítico)
A      V980.1          // e a tecla Enter for pressionad
a
S      V13.0, 2        // Seta o menu tanque Lub.
Redutores
R      V12.6, 2
R      V980.1, 1       // Reset a tecla Enter

NETWORK 22      //RETORNA AO MENU PRINCIPAL
LD      M0.3           // Tecla F4- voltar ao menu
principal
AN      M10.3          // Flag não está ativo
AN      M11.2          // Flag não está ativo
AN      M12.2          // Flag não está ativo
AN      M10.2          // Flag não está ativo
AN      M11.0          // Flag não está ativo
AN      M12.0          // Flag não está ativo
R      V12.0, 8        // Reset as mensagens
R      V13.0, 8        // Reset as mensagens
R      V14.0, 8        // Reset as mensagens
S      V14.0, 3        // Seta menu Principal
R      M0.0, 8         // Reset F4

NETWORK 23      // - PRIORIDADE 1- TANQUE LAMA - CRITICO
LD      V14.0
O      V14.1
O      V14.6           // Se Tanque Lub. Redutores nível
crítico
O      V14.7           // Se Tanque Lub. Redutores
nível de Alerta
O      V12.0           // Se Tanque de Preparação nível
crítico
O      V12.2           // Se Tanque de Preparação nível
de Alerta
O      V12.6           // Se Menu Principal ativo
O      V12.7           // Se Menu Principal ativo
O      V13.2           // Se Menu Principal ativo
O      V12.1
O      V13.3           // Se Tanque de Lama nível de
Alerta
O      M16.0           // Se Probe 1 defeito
O      M16.1           // Se Probe 2 defeito
O      M16.2           // Se Probe 3 defeito
A      V13.4           // E Tanque de Lama nível Critico
R      V12.0, 8        // Reset todas as mensagens
R      V13.0, 8        // Reset todas as mensagens

```

```

R      V14.0, 8           // Reset todas as mensagens
R      M16.0, 8           // Reset todas as mensagens
S      V13.4, 1           // Setar mensagem de estado
      critico Tanque de Lama

NETWORK 24           // - PRIORIDADE 2- TANQUE LAMA - ALERTA
LD     V14.0
O      V14.1
O      V12.2           // Se Tanque de Preparação nível
      de Alerta
O      V12.0           // Se Tanque de Preparação nível
      de Critico
O      V12.1
O      V14.7           // Se Tanque Lub. Redutores nível
      de Alerta
O      V14.6           // Se Tanque Lub. Redutores nível
      critico
O      M16.0           // Se Probe 1 defeito
O      M16.1           // Se Probe 2 defeito
O      M16.2           // Se Probe 3 defeito
A      V13.3           // E Tanque de Lama nível de
      Alerta
R      V12.0, 8         // Reset todas as mensagens
R      V13.0, 8         // Reset todas as mensagens
R      V14.0, 8         // Reset todas as mensagens
R      M16.0, 8         // Reset todas as mensagens
S      V13.3, 1         // Setar mensagem de estado de
      alerta Tanque de Lama
A      V580.1           // E pressionada a tecla ENTER
S      V13.0, 2         // Setar o Menu Tanque de Lama
R      V580.1, 1        // Reset o ENTER

NETWORK 25           // PRIORIDADE 3- TANQUE DE PREPARAÇÃO - CRITICO
LD     V14.6           // Se Tanque Lub. Redutores nível
      de Alerta
O      V14.7           // Se Tanque Lub. Redutores nível
      critico
O      V12.1           // Tanque de Lama abaixo de 10%
O      V12.0           // Se Tanque de Preparação nível
      de Alerta
O      M16.0           // Se Probe 1 defeito
O      M16.1           // Se Probe 2 defeito
O      M16.2           // Se Probe 3 defeito
A      V12.2           // Se Tanque de Preparação nível
      de Alerta
R      M16.0, 8         // Reset todas as mensagens
R      V14.6, 2         // Reset todas as mensagens
R      V12.0, 8         // Reset todas as mensagens
S      V12.2, 1         // Setar mensagem de Tan. de
      preparação nível critico

NETWORK 26           //PRIORIDADE 4- TANQUE DE PREPARAÇÃO NÍVEL - ALERTA
LD     V14.0
O      V14.1
O      V12.1
O      V14.6           // Se Tanque Lub. Redutores nível
      critico
O      V14.7           // Se Tanque Lub. Redutores nível
      de Alerta
O      M16.0           // Se Probe 1 defeito
O      M16.1           // Se Probe 2 defeito
O      M16.2           // Se Probe 3 defeito
O      V12.1           // Tanque de Lama abaixo de 10%
A      V12.0           // e o tanq. de preparação está
      em nível de alerta

```

```

R      V12.0, 8           // Reset todas as mensagens
R      V13.0, 8           // Reset todas as mensagens
R      V14.0, 8           // Reset todas as mensagens
R      M16.0, 8           // Reset todas as mensagens
S      V12.0, 1           // Seta mensagem de tanque de
Preparação Alerta
A      V580.1             // e ENTER for pressionado
S      V12.3, 3           // Setar o menu principal
R      V580.1, 1          // Reser o ENTER

NETWORK 27              // PRIORIDADDE 4- TANQUE LUB. REDUTORES
LD     V14.0
O      V14.1
O      V12.1
O      V14.6             // Se Tanque Lub. Redutores nível
de Alerta
O      M16.0             // Se Probe 1 defeito
O      M16.1             // Se Probe 2 defeito
O      M16.2             // Se Probe 3 defeito
O      V12.1             // Tanque de Lama abaixo de 10%
O      V12.6
O      V12.7
A      V14.7             // E Tanque Lub. Redutores nível
critico
R      M16.0, 3          // Reset todas as mensagens
R      V14.0, 8          // Reset todas as mensagens
R      V12.0, 8          // Reset todas as mensagens
R      V13.0, 8          // Reset todas as mensagens
R      M16.0, 8          // Reset todas as mensagens
S      V14.6, 1          // Setar a mensagem Tanque Lub.
Redutores nível de critico

```

